
  Unit -5
An OOP Application Framework

The Basic Directory Structure

Before we start creating content, let's lay out the basic directory structure for our application.

Follow these steps:

1. Create a directory named AppName on your hard drive. The AppName directory will contain everything in our project, including source code and final output. In a real project, you'd use your application's actual name instead of AppName.

2. In the AppName directory, create a subdirectory named deploy. The deploy directory will contain the final, compiled application, ready for posting to a web site or other distribution medium.

3. In the AppName directory, create a subdirectory named source. The source directory will contain all source code for the application, including classes (.as files) and Flash documents (.fla files).

Your application directory structure should look like this:

AppName/

   deploy/

   source/
The Flash Document (.fla file)

Every Flash application must include at least one Flash document (.fla file). The Flash document is the source file from which a Flash movie (.swf file) is exported. The Flash movie is what's actually displayed in the Flash Player (i.e., the Flash runtime environment). Many Flash applications are broken down into multiple .swf files that are loaded individually at runtime.

To create our application's main .fla file, follow these steps:

1. In the Flash authoring tool, choose File (New.

2. In the New Document dialog box, on the General tab, for Type, choose Flash Document, then click OK.

3. Use File (Save As to save the Flash document as AppName.fla in the AppName/source directory.
The Classes

Because we're building a pure OOP application, all of our application's content is created by its classes. In Flash, most applications are visual and include graphical user interfaces. Hence, the classes of an OOP Flash application typically create and manage user interface components. 

To create the directories for the com. some domain package, follow these steps:

1. In the AppName/source directory, create a subdirectory named com.

2. In the AppName/source/com directory, create a subdirectory named some domain.

To create class A, follow these steps if you're using Flash MX Professional 2004. If you're using Flash MX 2004 (standard edition), use an external text editor to create the .as class file following similar steps:

1. In Flash MX Professional 2004, choose File (New.

2. In the New Document dialog box, on the General tab, for Type, choose Action Script File.

3. Click OK. The script editor launches with an empty file.

4. Enter the following code into the script editor:

5. import com.somedomain.B;

6. class com.somedomain.A {

7.   private static var bInstance:B;

8.   public function A ( ) {

9.     // In this example, class A's constructor is not used.

10.   }

11.   public static function main ( ):Void {

12.     trace("Starting application.");

13.     bInstance = new B( );

14.   }

}

15. Choose File(Save As.

16. In the Save As dialog box, specify A.as as the filename, and save the file in the directory: AppName/source/com/somedomain.

Similarly, to create class B, use Flash MX Professional 2004 or an external text editor to create the B.as file in the AppName/source/com/somedomain folder with the following content:

class com.somedomain.B {

  public function B ( ) {

    trace("An instance of class B was constructed.");

  }

}

Notice that in our example framework we store the files for our class hierarchy under the AppName/source subdirectory.

The Document Timeline

We've now created two classes (A and B) and a Flash document that will use them (AppName.fla). Let's see exactly how AppName.fla loads the classes and invokes the class method A.main( ), which starts our application.

In our example application framework, we'll use the timeline of AppName.fla to create two basic application states: loading and main. Each state is implemented by pausing the playhead in a corresponding labeled frame. On the frame labeled loading, we display a loading message while our application's classes load. On the frame labeled main, we invoke A.main( ), which starts the application.

To load our A and B classes, we'll follow these general steps:

1. Specify the export frame for classes in the movie.

2. Add the labeled state frames loading and main to AppName.fla's timeline.

3. Add code that displays a loading message while the movie loads.

The specific procedures for those three steps are listed next. If you've already read  in previous chapters you'll have seen some of these procedures before.

To specify the export frame for classes in the movie, follow these steps:

1. Open AppName.fla in the Flash authoring tool.

2. Choose File(Publish Settings.

3. In the Publish Settings dialog box, on the Flash tab, next to the ActionScript Version, click Settings.

4. In the ActionScript Settings dialog box, for the Export Frame for Classes option, enter 10. The choice of frame 10 is arbitrary, but the specified frame must come after the code that displays the loading message (i.e., after the preloader).

5. Click OK to confirm the ActionScript settings.

6. Click OK to confirm the publish settings.

To add the labeled state frames loading and main to AppName.fla's timeline, follow these steps:

1. In AppName.fla's main timeline, double-click Layer 1 and rename it to scripts. We'll place all our code on the scripts layer. As a best practice, you should keep all your timeline code in the scripts layer, which should be the topmost layer in your timeline. The scripts layer should contain only scripts, not movie clips or other content. Some developers call the layer actions instead of scripts, but the premise is the same.

2. In the main timeline of AppName.fla, select frame 15 of the scripts layer.

3. Choose Insert (Timeline(Keyframe (F6).

4. Choose Insert(Timeline(Layer.

5. Double-click the new layer's name and change it to labels.

6. At frames 4 and 15 of the labels layer, add a new keyframe (Insert(Timeline (Keyframe or F6). Just as the scripts layer holds all our scripts, the labels layer is used exclusively to hold frame labels. Frame labels are a convenient, human-friendly way to refer to frames instead of referring to frames by number.

7. With frame 4 of the labels layer selected, in the Properties panel, under Frame, change <Frame Label> to loading.

8. With frame 15 of the labels layer selected, in the Properties panel, under Frame, change <Frame Label> to main.

Next, follow these steps to add code that displays a loading message while the movie loads:

1. At frame 5 of the scripts layer, add a new keyframe (Insert(Timeline(Keyframe or F6).

2. With frame 5 of the scripts layer selected, enter the following code into the Actions panel (F9):

3. if (_framesloaded == _totalframes) {

4.   gotoAndStop("main");

5. } else {

6.   gotoAndPlay("loading");

}

7. With frame 1 of the scripts layer selected, enter the following code into the Actions panel:

8. this.createTextField("loadmsg_txt", 0, 200, 200, 0, 0);

9. loadmsg_txt.autoSize = true;

loadmsg_txt.text = "Loading...Please wait.";

10. With frame 15 of the scripts layer selected, enter the following code into the Actions panel:


loadmsg_txt.removeTextField( );

We've now provided the basic timeline structure that loads our application's classes. All that's left is to start the application by invoking A.main( ). We do that on the frame labeled main in AppName.fla. Add the following code to the end of the script on frame 15 (i.e., just below loadmsg_txt.removeTextField( );):

import com.somedomain.A;

A.main( );
The Exported Flash Movie (.swf file)

Our application is now ready for testing and—assuming all goes well—deployment. To specify the directory in which to create AppName.swf, follow these steps:

1. With AppName.fla open, choose File (Publish Settings( Formats.

2. Under the File heading, for Flash (.swf), enter ../deploy/AppName.swf.

3. Click OK.

4. To test our application in the Flash authoring tool's Test Movie mode, select Control (Test Movie. Testing a movie actually creates AppName.swf in the AppName/deploy directory and immediately loads it into a debugging version of the Flash Player.

If your application is working, you should see the following appear in the Output panel:


Starting application.


An instance of class B was constructed.

When everything works in Test Movie mode, publish an HTML page that includes the movie, ready for posting to a web site as follows:

1. With AppName.fla open, choose File (Publish Settings( Formats.

2. Under the File heading, for HTML (.html), enter ../deploy/AppName.html.

3. Click Publish.

4. Click OK.

5. Test locally by opening AppName.html in your web browser.

6. When the local testing proves successful, upload the .html and .swf files to your web server.

7. Test in your browser from the remote site by browsing to the URL where you uploaded the .html file. Note that the package location and author-time folder structure do not matter once you upload your files to the server. Packages and classpaths matter only at compile time. Furthermore, you can upload the .html and .swf wherever you like on your server, but in our example, the two files must reside in the same web server folder.
Using Components with ActionScript 2.0
Currency Converter Application Overview

Our example GUI application is a simple currency converter. Figure 12-1 shows the components used in our currency converter interface (Button, ComboBox, Label, TextArea, and TextInput). The user must specify an amount in Canadian dollars, select a currency type from the drop-down list, and click the Convert button to determine the equivalent amount in the selected currency. The result is displayed on screen.

Figure 12-1. The currency converter application


We'll place the assets for our application in the following directory structure, which mirrors the structure we used in Chapter 11. Note that the deploy and source folders are both subdirectories of CurrencyConverter and that org/moock/tools is a subdirectory of the source folder:

CurrencyConverter/

  deploy/

  source/

    org/

      moock/

        tools/

Our application's main Flash document is named CurrencyConverter.fla. It resides in CurrencyConverter/source. To create the CurrencyConverter.fla file, we'll copy the file AppName.fla to the CurrencyConverter/source directory, and rename the file to CurrencyConverter.fla. That gives CurrencyConverter.fla the basic structure it needs, including a class preloader on frames 2 and 3 and a frame labeled main, on which we'll start the application.

Our application's only class is CurrencyConverter. It is stored in an external .as class file named CurrencyConverter.as, which resides in CurrencyConverter/source/org/moock/tools. Our exported application (a Flash movie) is named CurrencyConverter.swf. It resides in CurrencyConverter/deploy.

Adding Components to the Document

Here are the steps to follow to add the Button component to CurrencyConverter.fla's Library. To add the ComboBox, Label, TextArea, and TextInput components to the Library, follow the same steps, choosing the appropriate component instead of Button.

1. With CurrencyConverter.fla open in the Flash authoring tool, select Window (Development Panels ( Components.

2. In the Components panel, in the folder named UI Components, click and drag the Button component to the Stage. The Button component appears in the Library.

3. Select and delete the Button instance from the Stage.

To load our components after frame 10, we must set them to not export on frame 1, and then we must place a dummy instance of each component on stage after frame 10. The dummy instance is not used; it merely forces the component to load.

Here are the steps we follow to prevent the Button component from being exported on frame 1. To prevent the remaining components from being exported on frame 1, follow the same steps, choosing the appropriate component instead of Button.

1. Select the Button component in the Library.

2. From the Library's pop-up Options menu in the top-right corner of the panel, select Linkage.

3. In the Linkage Properties dialog box, under Linkage, uncheck the Export in First Frame checkbox.

4. Click OK.

When a component's Export in First Frame option is unchecked, the component is not compiled with the movie unless an instance of the component is placed on the document's timeline. The component loads at the frame on which an instance of it is placed. But the component initialization process requires a movie's ActionScript 2.0 classes to be available before the component is exported. Hence, in our CurrencyConverter.fla document, we'll place an instance of each of our components on frame 12, two frames after our document's ActionScript 2.0 classes are exported. To store the dummy component instances, we'll create a new timeline layer and keyframe, as follows:

1. Select Insert(Timeline(Layer.

2. Double-click the new layer and rename it load components.

3. Select frame 12 in the load components layer.

4. Select Insert(Timeline(Keyframe.

5. Select frame 13 in the load components layer.

6. Select Insert(Timeline(Keyframe. This second keyframe prevents the dummy component instances from showing up in our application. We need them only for loading; the CurrencyConverter class handles the actual creation of component instances in our application.

With our component-loading keyframe prepared, we can now place a dummy instance of each component on our document's timeline, as follows:

1. Select frame 12 in the load components layer.

2. From the Library, drag an instance of each component to the Stage.

3. Optionally use the Component Inspector panel (Window (Development Panels( Component Inspector) to add dummy text to each component instance indicating that it is not used, as shown in Figure 12-2. Consult Flash's online help for instructions on setting component parameters via the Component Inspector.

Figure 12-2 shows how our document's timeline and Stage look with frame 12 of the load components layer selected.

Figure 12-2. CurrencyConverter.fla's timeline and Stage

Starting the Application

import org.moock.tools.CurrencyConverter;

CurrencyConverter.main(this, 0, 150, 100);

Notice that the import statement allows us, in the future, to refer to the CurrencyConverter class by its name without typing its fully qualified package path. As implied by the preceding code, our class's main( ) method expects four parameters: the movie clip to hold the currency converter (this) and the depth, horizontal position, and vertical position—0, 150 and 100, respectively—at which to display the converter within the clip.

The CurrencyConverter Class

The CurrencyConverter class's three main duties are to:

· Provide a method that starts the application (main( ))

· Create the application interface

· Respond to user input

Example 12-1. The CurrencyConverter class

// Import the package containing the Flash UI components we're using.

import mx.controls.*;

// Define the CurrencyConverter class, and include the package path.

class org.moock.tools.CurrencyConverter {

  // Hardcode the exchange rates for this example.

  private static var rateUS:Number = 1.3205;  // Rate for US dollar

  private static var rateUK:Number = 2.1996;  // Rate for pound sterling

  private static var rateEU:Number = 1.5600;  // Rate for euro 

  // The container for all UI elements

  private var converter_mc:MovieClip;

  // The user interface components

  private var input:TextInput;           // Text field for original amount

  private var currencyPicker:ComboBox;   // Currency selection menu

  private var result:TextArea;           // Text field for conversion output

  public function CurrencyConverter (target:MovieClip, depth:Number,

                                     x:Number, y:Number) {

    buildConverter(target, depth, x, y);

  }

  /**

   * Creates the user interface for the currency converter

   */

  public function buildConverter (target:MovieClip, depth:Number,

                                  x:Number, y:Number):Void {

    // Store a reference to the current object for use by nested functions.

    var thisConverter:CurrencyConverter = this;

    // Make a container movie clip to hold the converter's UI.

    converter_mc = target.createEmptyMovieClip("converter", depth);

    converter_mc._x = x;

    converter_mc._y = y;

    // Create the title.

    var title:Label = converter_mc.createClassObject(Label, "title", 0);

    title.autoSize = "left";

    title.text = "Canadian Currency Converter";

    title.setStyle("color", 0x770000);

    title.setStyle("fontSize", 16);

    // Create the instructions.

    var instructions:Label = converter_mc.createClassObject(Label,

                                              "instructions",1);                                                      

    instructions.autoSize = "left";

    instructions.text = "Enter Amount in Canadian Dollars";

    instructions.move(instructions.x, title.y + title.height + 5);

    // Create an input text field to receive the amount to convert.

    input = converter_mc.createClassObject(TextInput, "input", 2);

    input.setSize(200, 25);

    input.move(input.x, instructions.y + instructions.height);

    input.restrict = "0-9.";

    // Handle this component's enter event using a generic listener object.

    var enterHandler:Object = new Object( );

    enterHandler.enter = function (e:Object):Void {

      thisConverter.convert( );

    }

    input.addEventListener("enter", enterHandler);

    // Create the currency selector ComboBox.

    currencyPicker = converter_mc.createClassObject(ComboBox, "picker", 3);

    currencyPicker.setSize(200, currencyPicker.height);

    currencyPicker.move(currencyPicker.x, input.y + input.height + 10);

    currencyPicker.dataProvider = [

                         {label:"Select Target Currency", data:null},

                         {label:"Canadian to U.S. Dollar", data:"US"},

                         {label:"Canadian to UK Pound Sterling", data:"UK"},

                         {label:"Canadian to EURO", data:"EU"}];

    // Create the Convert button.

    var convertButton:Button = converter_mc.createClassObject(Button,

                                                    "convertButton",4);

         convertButton.move(currencyPicker.x + currencyPicker.width + 5, 

                      currencyPicker.y);

    convertButton.label = "Convert!";

    // Handle this component's events using a handler function.

    // As discussed later under "Handling Component Events," this technique

    // is discouraged by Macromedia.

    convertButton.clickHandler = function (e:Object):Void {

      thisConverter.convert( );

    };

    // Create the result output field.

    result = converter_mc.createClassObject(TextArea, "result", 5);

    result.setSize(200, 25);

    result.move(result.x, currencyPicker.y + currencyPicker.height + 10);

    result.editable = false;

  }

  /**

   * Converts a user-supplied quantity from Canadian dollars to

   * the selected currency.

   */

  public function convert ( ):Void {

    var convertedAmount:Number;

    var origAmount:Number = parseFloat(input.text);

    if (!isNaN(origAmount)) {

      if (currencyPicker.selectedItem.data != null) {

        switch (currencyPicker.selectedItem.data) {

          case "US":

           convertedAmount = origAmount / CurrencyConverter.rateUS;

            break;

          case "UK":

            convertedAmount = origAmount / CurrencyConverter.rateUK;

            break;

          case "EU":

            convertedAmount = origAmount / CurrencyConverter.rateEU;

            break;

        }

        result.text = "Result: " + convertedAmount;

      } else {

        result.text = "Please select a currency.";

      }

    } else {

      result.text = "Original amount is not valid.";

    }

  }

  // Program point of entry

  public static function main (target:MovieClip, depth:Number,

                               x:Number, y:Number):Void {

    var converter:CurrencyConverter = new CurrencyConverter(target, depth,

                                                            x, y);

  }

}

Exporting the Final Application

Our currency converter application is ready for testing and deployment. To specify the name of the movie to create (CurrencyConverter.swf), follow these steps:

1. With CurrencyConverter.fla open, choose File (Publish Settings (Formats.

2. Under the File heading, for Flash (.swf), enter ../deploy/CurrencyConverter.swf.

3. Click OK.

4. To test our application in the Flash authoring tool's Test Movie mode, select Control (Test Movie.

MovieClip Subclasses
A movie clip is a self-contained multimedia object with a timeline for changing state (you can think of animation in a movie clip as a rapid succession of visual state changes). Movie clips can contain graphics, video, and audio.

Every MovieClip subclass has two parts: a movie clip symbol and a corresponding ActionScript 2.0 class. Like any other subclass, a MovieClip subclass uses the extends keyword to inherit from MovieClip:


class SomeClass extends MovieClip {


}

To create instances of a MovieClip subclass, we do not use the new operator as we would with a typical class. Instead, instances of the movie clip subclass's symbol (not the subclass itself) are created either manually in the authoring tool or programmatically via attachMovie( ) or duplicateMovieClip( ). It is, hence, not possible to create a functioning MovieClip subclass without a corresponding movie clip symbol in the Library of a .fla file. 
Our example MovieClip subclass, named Avatar, is an on-screen representation of a user in a chat room or a game.

The AvatarSymbol Movie Clip

As we learned earlier, every MovieClip subclass has a corresponding movie clip symbol. Our Avatar class's movie clip symbol is called AvatarSymbol. It contains a graphical depiction of the user in three different states: Idle, Sad, and Happy. The three states correspond to three labeled frames in AvatarSymbol's timeline. To change the state of the avatar, we position AvatarSymbol's playhead at the appropriate frame in its timeline. 

Figure 13-1. The AvatarSymbol timeline

Here are the steps we follow to create the AvatarSymbol movie clip, which we'll later associate with the Avatar class (see "Linking AvatarSymbol to the Avatar Class"):

1. Create a new Flash document named AvatarDemo.fla.

2. Select Insert (New Symbol.

3. For the symbol Name, enter AvatarSymbol. The name is arbitrary, but by convention, you should name the symbol ClassSymbol where Class is the name of the class with which you expect to associate the symbol.

4. For the symbol's Behavior type, select Movie Clip.

5. Click OK. Flash automatically enters Edit mode for the AvatarSymbol symbol.

Here are the steps we follow to build AvatarSymbol's timeline and contents:

1. Double-click Layer 1 and rename it to graphics.

2. Select frame 30 of the graphics layer.

3. Choose Insert (Timeline (Frame (F5).

4. Using Insert(Timeline (Layer, create two new layers, and name them labels and scripts.

5. With frame 1 of the scripts layer selected, enter the following code into the Actions panel (F9):



stop( );

6. Using Insert(Timeline (Blank Keyframe (F7), create blank keyframes on the labels and graphics layers at keyframes 10 and 20.

7. With frame 1 of the labels layer selected, in the Properties panel, under Frame, change <Frame Label> to IDLE.

8. With frame 10 of the labels layer selected, in the Properties panel, under Frame, change <Frame Label> to SAD.

9. With frame 20 of the labels layer selected, in the Properties panel, under Frame, change <Frame Label> to HAPPY.

10. On the graphics layer, use Flash's drawing tools to draw an idle face on frame 1, a sad face on frame 10, and a happy face on frame 20.

You may have noticed that we added a stop( ) statement on frame 1 but not on frames 10 and 20. We'll see why in a moment. Now that the AvatarSymbol symbol is complete, let's turn our attention to the Avatar class.

The Avatar Class

The basic code is:

class Avatar extends MovieClip {

}

The Avatar class defines two instance methods:

init( ) 

Initializes each Avatar instance

setState( ) 

Sets the avatar's state using one of three constants (class properties)—IDLE, SAD, and HAPPY.

Example 13-1. Avatar, a MovieClip subclass

class Avatar extends MovieClip {

  public static var HAPPY:Number = 0;

  public static var SAD:Number = 1;

  public static var IDLE:Number = 2;

  public function init ( ):Void {

    setState(Avatar.HAPPY);

  }

 public function setState(newState:Number):Void {

    switch (newState) {

      case Avatar.HAPPY:

        this.gotoAndStop("HAPPY");

       break;

      case Avatar.SAD:

        this.gotoAndStop("SAD");

        break;

      case Avatar.IDLE:

        this.gotoAndStop("IDLE");

        break;

    }

  }

}

Linking AvatarSymbol to the Avatar Class

To associate the AvatarSymbol movie clip with the Avatar class, we must set the AvatarSymbol's so-called "AS 2.0 Class" in the Flash authoring tool, as follows:

1. Select the AvatarSymbol movie clip in AvatarDemo.fla's Library.

2. Select the pop-up Options menu in the top-right corner of the Library panel, and choose the Linkage option.

3. In the Linkage Properties dialog box, for Linkage, select Export for ActionScript.

4. In the Linkage Properties dialog box, for Identifier, enter AvatarSymbol.

5. In the Linkage Properties dialog box, for AS 2.0 Class, enter Avatar.

6. Click OK.

Creating Avatar Instances

We create an instance of a MovieClip subclass by creating an instance of the associated library symbol using attachMovie( ) (exactly as we create a normal MovieClip instance). Here's the code:

var av:Avatar = Avatar(someMovieClip.attachMovie("AvatarSymbol", 

                                                 "avatar", 0));

Note that the preceding cast is an (unsafe) downcast. If we were being more cautious, we'd use:

var av:Avatar;

var temp_mc:MovieClip = someMovieClip.attachMovie("AvatarSymbol", 

                                                  "avatar", 0);

// Verify that the cast will succeed at runtime before attempting it.

if (temp_mc instanceof Avatar) {

  av = Avatar(temp_mc);

} else {

  trace("Warning: could not cast temp_mc to type Avatar.")

}
Initializing Avatar Instances

To initialize our new Avatar instance, we invoke init( ) on it, as follows:


av.init( );

Creating Avatar instances with the attachMovie( ) syntax can be ungainly because we have to specify the AvatarSymbol and the name of that symbol's physical incarnation ("avatar") in the call. Furthermore, requiring developers to call init( ) on each Avatar is sure to lead to errors. To smooth the instance-creation process, we'll create a class method, createAvatar( ). The createAvatar( ) method creates and initializes a new Avatar instance, wrapping the attachMovie( ) and init( ) calls into a single call for the end developer. Here's the code:

public static function createAvatar (name:String, target:MovieClip, 

                                     depth:Number):Avatar {

  var av:Avatar = Avatar(target.attachMovie("AvatarSymbol", name, depth));

  av.init( );

  return av;

}

The createAvatar( ) method conveniently supplies the name of the symbol associated with the Avatar class, so users of the class don't need to know or remember that it's AvatarSymbol. The method also invokes init( ) on the new instance after it's created, so we can be sure init( ) will always be invoked on new Avatar instances. Here's how we'd use the createAvatar( ) method:

var av:Avatar = Avatar.createAvatar("avatar", someMovieClip, 0);

The createAvatar( ) method also gives us the flexibility to define parameters for use during instance initialization, exactly like a constructor function. For example, we could easily add parameters to createAvatar( ) that set the new instance's initial position:

public static function createAvatar (name:String, target:MovieClip, 

                                     depth:Number, x:Number,

                                     y:Number):Avatar {

  var av:Avatar = Avatar(target.attachMovie("AvatarSymbol", name, depth));

  av.init(x, y);

  return av;

}

public function init (x:Number, y:Number):Void {

  setState(Avatar.HAPPY);

  this._x = x;

  this._y = y;

}

Our avatar-creation code would now look like this:

var av:Avatar = Avatar.createAvatar("avatar", someMovieClip, 0, 100, 100);

which is much tidier and easier to understand than:

var av:Avatar = someMovieClip.attachMovie("AvatarSymbol", "avatar", 0);

av.init(100, 100);

Here's the code for the completed Avatar class, showing the new createAvatar( ) method and updated init( ) method in context with the rest of the class:

class Avatar extends MovieClip {

  public static var HAPPY:Number = 0;

  public static var SAD:Number = 1;

  public static var IDLE:Number = 2;

  public static function createAvatar (name:String, target:MovieClip, 

                                       depth:Number, x:Number,

                                       y:Number):Avatar {

    var av:Avatar = Avatar(target.attachMovie("AvatarSymbol", name, depth));

    av.init(x, y);

    return av;

  }

  public function init (x:Number, y:Number):Void {

    setState(Avatar.HAPPY);

    this._x = x;

    this._y = y;

  }

  public function setState(newState:Number):Void {

    switch (newState) {

      case Avatar.HAPPY:

        this.gotoAndStop("HAPPY");

        break;

      case Avatar.SAD:

        this.gotoAndStop("SAD");

        break;

      case Avatar.IDLE:

        this.gotoAndStop("IDLE");

        break;

    }

  }

}

Example 13-2 shows a revised version of the Avatar class. This version defines a constructor because, now that Avatar is no longer a subclass of MovieClip, instances are instantiated with new. The AvatarSymbol clip instance is created in the Avatar constructor, which replaces the createAvatar( ) method. Because initialization is usually performed in a class's constructor function, we put some of the code from init( ) in the constructor and move the rest to a new method, setPosition( ).

Example 13-2. The composition-based Avatar class

class Avatar {

  public static var HAPPY:Number = 0;

  public static var SAD:Number = 1;

  public static var IDLE:Number = 2;

  private var av_mc:MovieClip;

  public function Avatar (name:String, target:MovieClip, 

                          depth:Number, x:Number, y:Number) {

    av_mc = target.attachMovie("AvatarSymbol", name, depth);

    setState(Avatar.HAPPY);

    setPosition(x, y);

  }

  public function setState(newState:Number):Void {

    switch (newState) {

      case Avatar.HAPPY:

        av_mc.gotoAndStop("HAPPY");

        break;

      case Avatar.SAD:

        av_mc.gotoAndStop("SAD");

        break;

      case Avatar.IDLE:

        av_mc.gotoAndStop("IDLE");

        break;

    }

  }

  public function setPosition (x:Number, y:Number):Void {

    av_mc._x = x;

    av_mc._y = y;

  }

}

In many ways, the composition version of the Avatar class is cleaner and more flexible than its MovieClip subclass counterpart—cleaner because it uses fewer methods and a traditional constructor, more flexible because it can easily change the av_mc to some other movie clip symbol if required. However, the MovieClip subclass version would still be preferable in the following situations:

· When Avatar instances need to belong to the MovieClip datatype (e.g., for the sake of polymorphism)

· When the Avatar class and the AvatarSymbol movie clip symbol need to be packaged together as a component for distribution

· When Avatar instances need most of the methods of the MovieClip class (in which case the "Is-A" relationship is legitimate) 

Issues with Nested Assets

Movie clip symbols can contain other nested movie clips, text fields, or components (i.e., elements attached to the clip symbol during authoring or at runtime). To access and use these nested assets, we must sidestep two issues.

(1) Properties and Methods of Nested Assets Initially Undefined

The Flash Player takes an "outside-in" approach to component and movie clip initialization. When initializing a movie clip hierarchy in a movie, the Flash Player starts with the outermost clip in the hierarchy, defines its custom properties and methods, then moves on to its child clips, defines their custom properties and methods, and so on. As a result, immediately after a clip is constructed, it can see its parent's custom methods and properties, but it cannot see its children's custom methods and properties.

For example, consider a movie clip symbol, ChatRoomSymbol, that represents a chat room. It contains the following component instances:

outgoing 

A TextInput component for outgoing messages

incoming 

A TextArea component for incoming messages

userList 

A List component for the list of users in the room

Suppose we use a ChatRoomSymbol instance via composition in a class, ChatRoom, as follows:

class ChatRoom {

  private static var chatroomID:Number = 0;

  private var chat_mc:MovieClip;

  public function ChatRoom (target:MovieClip, depth:Number) {

    // Create clip instance.

    chat_mc = target.attachMovie("ChatRoomSymbol", 

                                 "chatroom" + ChatRoom.chatroomID++, 

                                 depth);

   chat_mc.userList.dataProvider = [{label:"Colin", data:"User1"},

                 {label:"Derek", data:"User2"},

                 {label:"James", data:"User3"}];

  }

}

The ChatRoom constructor creates an instance of the ChatRoomSymbol movie clip symbol and stores it in the property chat_mc. (By now, you should know that it actually stores the ActionScript object representation of the ChatRoomSymbol instance, so I'll no longer make the distinction.) The constructor then attempts to populate the nested userList List component by assigning a value to its dataProvider property:

chat_mc.userList.dataProvider = [{label:"Colin", data:"User1"},

             {label:"Derek", data:"User2"},

             {label:"James", data:"User3"}];

But the assignment has no effect because the userList component's properties and methods are not yet functional! To overcome this problem, we must use setInterval( ) to poll the userList component until its custom properties and methods are defined. Once we detect the existence of a single userList method, we can be sure that all of its properties and methods are available. In fact, we can also be sure that every other nested component is, likewise, ready for use. By the time a single frame passes, the custom properties and methods of all nested assets will be initialized.

The following new code for the ChatRoom class shows how to poll for the existence of a userList List component instance's method. The new code uses an init( ) method to initialize nested components. The init( ) method is called by the function passed to setInterval( ) as soon as the userList.addItem( ) method is defined. Notice that the function passed to setInterval( ) is, itself, passed a reference to the current object, this. This technique allows the function passed to setInterval( ) to invoke the init( ) method on the current object. Without the reference to this, the nested function called by setInterval( ) would have no access to the current object and could not invoke init( ).

class ChatRoom {

  private static var chatroomID:Number = 0;

  private var chat_mc:MovieClip;

  public function ChatRoom (target:MovieClip, depth:Number) {

    // Create clip instance.

    chat_mc = target.attachMovie("ChatRoomSymbol", 

                                 "chatroom" + ChatRoom.chatroomID++, 

                                 depth);

    // Wait until nested clips are initialized before
    // performing any operations on them.

    var initInt = setInterval(function (cr:ChatRoom):Void {
      if (cr.chat_mc.userList.addItem != undefined) {
        cr.init( );
        clearInterval(initInt);
      }
    }, 10, this);
  }

  public function init ( ):Void {
    // Hardcode some dummy data for this example.

    setUserList([{label:"Colin", data:"User1"},

                 {label:"Derek", data:"User2"},

                 {label:"James", data:"User3"}]);
  }
  public function setUserList (list:Array):Void {
    chat_mc.userList.dataProvider = list;
  }
}

(2) Nested Assets Not Automatically Recognized by Compiler

Assets nested in a movie clip symbol are not automatically available to a corresponding MovieClip subclass as instance properties. That is, a MovieClip subclass cannot refer to assets nested inside its associated movie clip symbol unless the subclass explicitly declares those assets as instance properties.

For example, consider a LoginSymbol movie clip symbol that contains two text fields, userName and password. The LoginSymbol's corresponding MovieClip subclass is Login. The Login class wants to set the default contents of the userName and password text fields. If we use the following Login class code:

class Login extends MovieClip {

  public function Login ( ) {

    userName.text = "Enter your name.";

    password.text = "Enter your password.";

  }

}

the compiler generates the following errors:

There is no property with the name 'userName'.

         userName.text = "Enter your name.";

There is no property with the name 'password'.

         password.text = "Enter your password.";

We know that the LoginSymbol movie clip symbol contains the required text fields, but the compiler doesn't have any way of knowing. The compiler bases its type checking on the Login class definition, not the contents of the LoginSymbol movie clip symbol. To work around this problem, we simply declare the userName and password text fields as instance properties, like this:

class Login extends MovieClip {

  public var userName:TextField;  // Declare userName
  public var password:TextField;  // Declare password
  public function Login ( ) {

    userName.text = "Enter your name.";

    password.text = "Enter your password.";

  }

}

A Note on MovieClip Sub-subclasses

Nothing officially prevents a MovieClip subclass from being subclassed. For example, hearkening back to the Avatar class, if we wanted to create an Avatar whose mood changed randomly, we could create a RandomAvatar class that inherits from Avatar. However, the RandomAvatar class would need to be associated with its own library movie clip symbol because, as we learned earlier, a movie clip symbol can specify only a single AS 2.0 Class association. Hence, while the RandomAvatar class would inherit the behavior of the Avatar class, it would not inherit Avatar's physical movie clip symbol. Hence, to create the RandomAvatar class, we'd have to create a completely separate symbol, RandomAvatarSymbol, that is a duplicate of AvatarSymbol.

Duplicating a symbol merely for the sake of subclassing is neither practical nor wise in most cases. The better approach is (yet again) to use composition. When a class uses a movie clip symbol via composition, the class can easily be subclassed, allowing reuse both of its code and of the movie clip symbol associated via composition. Example 13-3 shows how we could create the RandomAvatar class as a subclass of the Avatar class shown earlier in Example 13-2. Neither Avatar nor RandomAvatar are MovieClip subclasses. We use composition to associate them with a movie clip symbol.

Example 13-3. RandomAvatar, an Avatar subclass

class RandomAvatar extends Avatar {

  private var randomInt:Number;

  public function RandomAvatar (name:String, target:MovieClip, 

                          depth:Number, x:Number, y:Number) {

    super(name, target, depth, x, y);

    startRandom( );

  }

  public function startRandom ( ):Void {

    randomInt = setInterval (function (av:RandomAvatar):Void {

      var r:Number = Math.floor(Math.random( ) * 3);

      av.setState(r);

    }, 500, this);

  }

  public function stopRandom ( ):Void {

    clearInterval(randomInt);

  }

}

